/*
* Copyright 2011 Pavel Kozhin 
*
* Licensed under the Apache License, Version 2.0 (the "License"); 
* you may not use this file except in compliance with the License. 
* You may obtain a copy of the License at 
* 
* http://www.apache.org/licenses/LICENSE-2.0
* 
* Unless required by applicable law or agreed to in writing, software 
* distributed  under the License is distributed on an "AS IS" BASIS,
* WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied. 
* See the License for the specific language governing permissions and 
* limitations under the License.
*
* Author: Pavel Kozhin.
*/
package org.etherframework.controller
{
	import flash.errors.IllegalOperationError;
	
	import mx.rpc.IResponder;
	
	import raix.reactive.IObservable;
	import raix.reactive.Observable;
	
	public class ResponderCommand extends ReactiveCommand implements IResponderCommand
	{
		private var _responder:IResponder;
		
		public function respond(responder:IResponder):void
		{
			ensureIsInitialized();
			
			if (_responder)
				throw new IllegalOperationError("Command "+ this + " already has responder.");
			
			_responder = responder;
			execute();
		}
		
		override protected function errorHandler(error:Error):void
		{
			fault(error);
		}
		
		protected function result(value:*):IObservable
		{
			if (_responder && (_responder.result is Function))
			{
				_responder.result(value);
			}
			return Observable.value(value);
		}
		
		protected function fault(value:*):IObservable
		{
			if (_responder && (_responder.fault is Function))
			{
				_responder.fault(value);
			}
			return Observable.value(value);
		}
	}
}